In this Maple file, we compare our Lax pair and Hamiltonian evolutions with the one given by H. Chiba in "Multi-poisson approach to the Painlevé equations: from the isospectral deformation to the isomonodromic deformation".
Note that we place ourselves directly after the symplectic reduction. In particular \check{q}=q and \check{p}=p.

Loading the Lax matrices obtained by H. Chiba

```
> restart:
  with(LinearAlgebra):

  Sigma3:=Matrix(2,2,0):
  Sigma3[1,1]:=1:
  Sigma3[2,2]:=-1:
  Sigma3;

  LChiba:=Matrix(2,2,0):
  LChiba[1,1]:=lambda^3+lambda*(2*t2-1/2*V2)+t1-1/2*V3-1/2*V2*W1:
  LChiba[1,2]:=V2*lambda^2+V3*lambda+alpha4+t2*V2-1/4*V2^2-V2*
  tdW2-V3*W1:
  LChiba[2,1]:=lambda^2+lambda*W1+tdW2+t2:
  LChiba[2,2]:=-lambda^3-(2*t2-1/2*V2)*lambda-(t1-1/2*V3-1/2*V2*
  W1):
  LChiba;


  A1Chiba:=Matrix(2,2,0):
  A1Chiba[1,1]:=lambda-W1:
  A1Chiba[1,2]:=V2:
  A1Chiba[2,1]:=1:
  A1Chiba[2,2]:=-lambda+W1:
  A1Chiba;

  A2Chiba:=Matrix(2,2,0):
  A2Chiba[1,1]:=lambda^2+t2-1/2*V2-tdW2:
  A2Chiba[1,2]:=lambda*V2+V3:
  A2Chiba[2,1]:=lambda+W1:
  A2Chiba[2,2]:=-lambda^2-(t2-1/2*V2-tdW2):
  A2Chiba;
```

$$\begin{bmatrix} 1 & 0 \\ 0 & -1 \end{bmatrix} \qquad (1.1)$$

$$\left[\left[\lambda^3+\lambda\left(2\,t2-\frac{1}{2}\,V2\right)+t1-\frac{1}{2}\,V3-\frac{1}{2}\,V2\,W1,\ V2\,\lambda^2+V3\,\lambda+\alpha 4+t2\,V2-\frac{1}{4}\,V2^2-V2\,tdW2-V3\,W1\right],\right.$$

$$\left.\left[W1\,\lambda+\lambda^2+t2+tdW2,\ -\lambda^3-\lambda\left(2\,t2-\frac{1}{2}\,V2\right)-t1+\frac{1}{2}\,V3+\frac{1}{2}\,V2\,W1\right]\right]$$

$$\begin{bmatrix}\lambda-W1 & V2\\ 1 & -\lambda+W1\end{bmatrix}$$

$$\begin{bmatrix}\lambda^2+t2-\frac{1}{2}\,V2-tdW2 & V2\,\lambda+V3\\ \lambda+W1 & -\lambda^2-t2+\frac{1}{2}\,V2+tdW2\end{bmatrix}$$

Loading our Lax matrices \td{L} and \td{A}_1 and \td{A}_2

```
> tdL:=Matrix(2,2,0):
tdL[1,1]:=lambda^3+(-q1^3+q2^3+p1-p2)*lambda/(q1-q2)+(q1^3*q2+
(-q2^3+p2)*q1-p1*q2)/(q1-q2):
tdL[1,2]:=lambda^2+(-q1-q2)*lambda+q1*q2:
tdL[2,1]:= (tau2-(2*(p1-p2))/(q1-q2)+2*q1^2+2*q1*q2+2*q2^2)*
lambda^2+((q1+q2)*tau2+tau1-(2*(p1*q1-p2*q2))/(q1-q2)+(2*(q1+
q2))*(q1^2+q2^2))*lambda+(q1+q2)*tau1+(1/4)*tau2^2+(q1^2+q1*q2+
q2^2)*tau2-(p1-p2)^2/(q1-q2)^2+(2*(p1*q2-p2*q1))*(q1+q2)/(q1-
q2)+q1^4+q2^4-q1^2*q2^2-2*h+2*tinfty10:
tdL[2,2]:=-tdL[1,1]:
tdL;

tdA1:=Matrix(2,2,0):
tdA1[1,1]:=(1/2)*lambda+(1/2)*q1+(1/2)*q2:
tdA1[1,2]:=1/2:
tdA1[2,1]:=((q1-q2)*tau2+2*q1^3-2*q2^3-2*p1+2*p2)/(2*q1-2*q2):
tdA1[2,2]:=-(1/2)*lambda-(1/2)*q1-(1/2)*q2:
tdA1;

tdA2:=Matrix(2,2,0):
tdA2[1,1]:=(q1-q2)*lambda^2/(4*q1-4*q2)+(-q1^3-q1^2*q2+q1*q2^2+
q2^3+p1-p2)/(4*q1-4*q2):
tdA2[1,2]:=(1/4)*lambda-(1/4)*q1-(1/4)*q2:
tdA2[2,1]:=(1/2)*((p2-p1)/(q1-q2)+q1^2+q1*q2+q2^2+(1/2)*tau2)*
lambda+(1/4)*(q1+q2)*tau2+(1/4)*tau1+(-p1*q1+p2*q2)/(2*(q1-q2))
+(1/2)*(q1+q2)*(q1^2+q2^2):
tdA2[2,2]:=-tdA2[1,1]:
```

```
tdA2;
```

$$\left[\left[\lambda^3+\frac{(-q1^3+q2^3+p1-p2)\lambda}{q1-q2}+\frac{q1^3 q2+(-q2^3+p2) q1-p1 q2}{q1-q2}, \lambda^2+(-q1-q2)\lambda+q1 q2\right],\right. \tag{2.1}$$

$$\left[\left(\tau2-\frac{2(p1-p2)}{q1-q2}+2 q1^2+2 q1 q2+2 q2^2\right)\lambda^2+\left((q1+q2)\tau2+\tau1-\frac{2(p1 q1-p2 q2)}{q1-q2}+2(q1+q2)(q1^2+q2^2)\right)\lambda+(q1+q2)\tau1+\frac{1}{4}\tau2^2+(q1^2+q1 q2+q2^2)\tau2-\frac{(p1-p2)^2}{(q1-q2)^2}+\frac{2(p1 q2-p2 q1)(q1+q2)}{q1-q2}+q1^4+q2^4-q1^2 q2^2-2h+2\,tinfty10, -\lambda^3-\frac{(-q1^3+q2^3+p1-p2)\lambda}{q1-q2}-\frac{q1^3 q2+(-q2^3+p2) q1-p1 q2}{q1-q2}\right]\Big]$$

$$\begin{bmatrix} \frac{1}{2}\lambda+\frac{1}{2}q1+\frac{1}{2}q2 & \frac{1}{2} \\ \frac{(q1-q2)\tau2+2 q1^3-2 q2^3-2 p1+2 p2}{2 q1-2 q2} & -\frac{1}{2}\lambda-\frac{1}{2}q1-\frac{1}{2}q2 \end{bmatrix}$$

$$\left[\left[\frac{(q1-q2)\lambda^2}{4 q1-4 q2}+\frac{-q1^3-q1^2 q2+q1 q2^2+q2^3+p1-p2}{4 q1-4 q2}, \frac{1}{4}\lambda-\frac{1}{4}q1-\frac{1}{4}q2\right],\right.$$

$$\left[\frac{1}{2}\left(\frac{p2-p1}{q1-q2}+q1^2+q1 q2+q2^2+\frac{1}{2}\tau2\right)\lambda+\frac{1}{4}(q1+q2)\tau2+\frac{1}{4}\tau1+\frac{-p1 q1+p2 q2}{2 q1-2 q2}+\frac{1}{2}(q1+q2)(q1^2+q2^2), -\frac{(q1-q2)\lambda^2}{4 q1-4 q2}-\frac{-q1^3-q1^2 q2+q1 q2^2+q2^3+p1-p2}{4 q1-4 q2}\right]\Big]$$

Identifying the Lax matrices

We must first rescale the times to match both formalisms. This implies also to renormalize the auxiliary matrices accordingly.

```
> tau1:=2*t1:
  tau2:=4*t2:
  tdA1norm:=2*tdA1:
  tdA2norm:=4*tdA2:
```

We shall prove that we may identigy Chiba's coordinates with ours so that Chiba's Lax matrices are the transposed of ours.

```
> V2:=-2*(p1-p2)/(q1-q2)+2*q1^2+2*q1*q2+2*q2^2+4*t2;
  V3:=-2*(p1*q1-p2*q2)/(q1-q2)+2*q1^3+2*q1^2*q2+2*q1*q2^2+2*
  q2^3+4*q1*t2+4*q2*t2+2*t1;
```

```
W1:=-(q1+q2);
tdW2:=q1*q2-t2;
alpha4:=2*tinfty10-2*h;
simplify((tdL-Transpose(LChiba)));
simplify((tdA1norm-Transpose(A1Chiba)));
simplify((tdA2norm-Transpose(A2Chiba)));
```

$$V2 := -\frac{2\,(p1-p2)}{q1-q2}+2\,q1^2+2\,q1\,q2+2\,q2^2+4\,t2 \tag{3.1}$$

$$V3 := -\frac{2\,(p1\,q1-p2\,q2)}{q1-q2}+2\,q1^3+2\,q1^2\,q2+2\,q1\,q2^2+2\,q2^3+4\,q1\,t2+4\,q2\,t2+2\,t1$$

$$W1 := -q1-q2$$

$$tdW2 := q1\,q2-t2$$

$$\alpha 4 := 2\,tinfty10-2\,h$$

$$\begin{bmatrix} 0 & 0 \\ 0 & 0 \end{bmatrix}$$

$$\begin{bmatrix} 0 & 0 \\ 0 & 0 \end{bmatrix}$$

$$\begin{bmatrix} 0 & 0 \\ 0 & 0 \end{bmatrix}$$